package {
	
	import br.com.damazzy.manterTipoProduto.TipoProdutoEdicao;
	import br.com.damazzy.manterTipoProduto.TipoProdutoSelecaoView;
	import br.com.damazzy.util.TratamentoErros;
	import br.com.damazzy.vo.TipoProdutoVO;
	
	import flash.events.Event;
	import flash.events.MouseEvent;
	
	import mx.collections.ArrayCollection;
	import mx.controls.Alert;
	import mx.events.CloseEvent;
	import mx.events.FlexEvent;
	import mx.managers.PopUpManager;
	import mx.rpc.events.FaultEvent;
	import mx.rpc.events.ResultEvent;
	
	public class TipoProdutoSelecao extends TipoProdutoSelecaoView {
		
		private var tipoProdutoEdicao: TipoProdutoEdicao;
		private var listaTiposProduto: ArrayCollection = new ArrayCollection();
		
		public function TipoProdutoSelecao() {
			super();
			
			this.addEventListener(FlexEvent.CREATION_COMPLETE, init);
			
			include 'br/com/damazzy/util/RegistroVOs.as';
		}
		
		private function init(event: FlexEvent): void {
			
			btnPesquisar.addEventListener(MouseEvent.CLICK, pesquisar);
			btnIncluir.addEventListener(MouseEvent.CLICK, incluir);
			btnAlterar.addEventListener(MouseEvent.CLICK, alterar);
			btnExcluir.addEventListener(MouseEvent.CLICK, alertaExcluir);
			btnFechar.addEventListener(MouseEvent.CLICK, fechar);
			this.addEventListener(CloseEvent.CLOSE, fechar);
			
			rm.pesquisar.addEventListener(ResultEvent.RESULT, resultPesquisar);
			rm.pesquisar.addEventListener(FaultEvent.FAULT, TratamentoErros.fault);
			rm.excluir.addEventListener(ResultEvent.RESULT, resultExcluir);
			rm.excluir.addEventListener(FaultEvent.FAULT, TratamentoErros.fault);
			
			limparDados();
		}
		
		public function pesquisar(event: MouseEvent = null): void {
			
			var tipoProduto: TipoProdutoVO = new TipoProdutoVO();
			tipoProduto.descricao = txtDescricao.actualText == "" ? null : txtDescricao.actualText;
			
			rm.pesquisar(tipoProduto);
		}
		
		private function resultPesquisar(event: ResultEvent): void {
			listaTiposProduto = event.result.dados.lista;
			grdDados.dataProvider = listaTiposProduto;
			
			if (listaTiposProduto.length == 0) {
				Alert.show("Nenhum tipo de produto encontrado com o filtro informado.", "Aviso!");
			}
		}
		
		private function incluir(event: MouseEvent): void {
			tipoProdutoEdicao = new TipoProdutoEdicao(null);
			tipoProdutoEdicao.telaPai = this;
			PopUpManager.addPopUp(tipoProdutoEdicao, this, true);
			PopUpManager.centerPopUp(tipoProdutoEdicao);
		}
		
		private function alterar(event: MouseEvent): void {
			
			if (grdDados.selectedItem != null) {
				
				tipoProdutoEdicao = new TipoProdutoEdicao(grdDados.selectedItem as TipoProdutoVO);
				tipoProdutoEdicao.telaPai = this;
				PopUpManager.addPopUp(tipoProdutoEdicao, this, true);
				PopUpManager.centerPopUp(tipoProdutoEdicao);
				
			} else {
				Alert.show("Selecione um tipo de produto na lista para alterar.", "Aviso!");
			}
		}
		
		private function alertaExcluir(event: MouseEvent): void {
			if (grdDados.selectedItem != null) {
				Alert.yesLabel = "Sim";
                Alert.noLabel = "Não";
				Alert.show("Deseja realmente excluir o tipo de produto selecionado?", "Exclusão", 1|2, this, excluir);
			} else {
				Alert.show("Selecione um tipo de produto na lista e clique em excluir.", "Aviso!");
			}
		}
		
		private function excluir(event: CloseEvent): void {
			if (event.detail == Alert.YES) {
				var tipoProdutoExcluido: TipoProdutoVO = grdDados.selectedItem as TipoProdutoVO;
				rm.excluir(tipoProdutoExcluido);
			}
		}
		
		private function resultExcluir(result: ResultEvent): void {
			Alert.show("Tipo de produto excluído com sucesso.", "Aviso!");
			pesquisar();
		}
		
		private function fechar(event: Event): void {
			PopUpManager.removePopUp(this);
		}
		
		private function limparDados(): void {
			txtDescricao.actualText = "";
			
			grdDados.dataProvider = null;
		}
	}
}